3. ESTYMACJA PRZEDZIAŁOWA

3.1. Przedziały ufności

Lp.	Parametr	Uwagi	Przedział ufności	Parametry
1.	μ	znane σ	$\left[\bar{x}-\varepsilon\frac{\sigma}{\sqrt{n}},\ \bar{x}+\varepsilon\frac{\sigma}{\sqrt{n}}\right],$	$\varepsilon=-\Phi^{-1}\left(\frac{\alpha}{2}\right)$
2.	μ	nieznane σ duża próba	*jak w 1.*	$\sigma\approx S$ lub $\sigma\approx s$
3.	μ	nieznane σ mała próba	$\left[\bar{x}-\varepsilon\frac{s}{\sqrt{n}},\ \bar{x}+\varepsilon\frac{s}{\sqrt{n}}\right]$	$\varepsilon=-F^{-1}_{t(n-1)}\left(\frac{\alpha}{2}\right)$
4.	σ	duża próba	$\left[\frac{s}{1+\varepsilon/\sqrt{2n}},\ \frac{s}{1-\varepsilon/\sqrt{2n}}\right]$	ε *jak w 1.*
5.	σ^2	duża próba	$\left[\left(\frac{s}{1+\varepsilon/\sqrt{2n}}\right)^2,\ \left(\frac{s}{1-\varepsilon/\sqrt{2n}}\right)^2\right]$	ε *jak w 1.*
6.	σ	mała próba	$\left[\sqrt{\frac{(n-1)s^2}{\varepsilon_2}},\ \sqrt{\frac{(n-1)s^2}{\varepsilon_1}}\right]$	$\varepsilon_1=F^{-1}_{\chi^2(n-1)}\left(\frac{\alpha}{2}\right),$ $\varepsilon_2=F^{-1}_{\chi^2(n-1)}\left(1-\frac{\alpha}{2}\right)$
7.	σ^2	mała próba	$\left[\frac{(n-1)s^2}{\varepsilon_2},\ \frac{(n-1)s^2}{\varepsilon_1}\right]$	ε_1 i ε_2 *jak w 6.*
8.	p	duża próba	$\left[\hat{p}-\varepsilon\sqrt{\frac{\hat{p}(1-\hat{p})}{n}},\ \hat{p}+\varepsilon\sqrt{\frac{\hat{p}(1-\hat{p})}{n}}\right]$	$\hat{p}=\frac{m}{n}$, ε *jak w 1.*

$$s=\sqrt{\frac{1}{n-1}\sum_{i=1}^{n}(x_i-\bar{x})^2}\ ,\ S=\sqrt{\frac{1}{n}\sum_{i=1}^{n}(x_i-\bar{x})^2}\ .$$

3.2. Przedziały ufności w Excel-u

μ (znane σ)

UFNOŚĆ(alfa; odchylenie_std; wielkość)

gdzie:

alfa – poziom istotności; **odchylenie_std** – odchylenie standardowe rozkładu; **wielkość** – rozmiar próby.
Uwaga! Pozostałe przedziały ufności wyznacza się z wzorów (patrz punkt 3.1/).

Przykład 1.

Na podstawie zebranych w poniższej tabeli pomiarów długości pewnego detalu wykonanych z dokładnością $\sigma = 1.5$ oszacować na *poziomie ufności* $(1 - \alpha) = 0.99$ rzeczywistą długość tego detalu.

21.9	17.8	23	21.3	19.4	18	21	22.4	21	17.6

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n**	10									
3											
4	**poziom ufności (1 – α)**	0,99									
5	**odchylenie standardowe σ**	1,5									
6	**α**	**=1-B4**									
7	**ε**	**=-ROZKŁAD.NORMALNY.S.ODW(B6/2)**									
8	**x_{sr}**	**=ŚREDNIA(B1:K1)**									
9	**d=ε σ/√n**	**=B7*B5/PIERWIASTEK(B2)**						**=UFNOŚĆ(B6;B5;B2)**			
10	**przedział ufności (lewa strona)**	**=B8-B9**									
11	**przedział ufności (prawa strona)**	**=B8+B9**									

Uwaga! W powyższym arkuszu pokazane zostały dwie alternatywne metody rozwiązania zadania.

Przykład 2.

Rozwiązać zadanie z przykładu 1. zakładając, że dokładność pomiarów nie jest znana.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n**	10									
3											
4	**poziom ufności (1 – α)**	0,99									
5	**α**	**=1-B4**									
6	**s**	**=ODCH.STANDARD.POPUL(B1:K1)**									
7	**x_{sr}**	**=ŚREDNIA(B1:K1)**									
8	**ε**	**=ROZKŁAD.T.ODW(B5;B2-1)**									
9	**d=ε s/√(n-1)**	**=B8*B6/PIERWIASTEK(B2-1)**									
10	**przedział ufności (lewa strona)**	**=B7-B9**									
11	**przedział ufności (prawa strona)**	**=B7+B9**									

Przykład 3.

Oszacować na poziomie ufności (1 – α) = 0.99 średni błąd kwadratowy σ pomiarów długości detalu z przykładu 1.

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n**	10									
3											
4	**poziom ufności (1 – α)**	0,99									
5	α	**=1-B4**									
6	ε1	**=ROZKŁAD.CHI.ODW(1-B5/2;B2-1)**									
7	ε2	**=ROZKŁAD.CHI.ODW(B5/2;B2-1)**									
8	s^2	**=WARIANCJA.POPUL(B1:K1)**									
9	**przedział ufności (lewa strona)**	**=PIERWIASTEK(B2*B8/B7)**									
10	**przedział ufności (prawa strona)**	**=PIERWIASTEK(B2*B8/B6)**									

Gdyby rozmiar próby był większy ($n \geq 30$), *przedział ufności* można byłoby wyznaczyć posługując się zmienną o rozkładzie $\mathcal{N}(0, 1)$.

	A	B	C	D	E	F	G	H	I	J	K
6	**s**	**=ODCH.STANDARD.POPUL(B1:K1)**									
7	ε	**=-ROZKŁAD.NORMALNY.S.ODW(B5/2)**									
8	ε /√(2n)	**=B7/PIERWIASTEK(2*B2)**									
9	**przedział ufności (lewa strona)**	**=B6/(1+B8)**									
10	**przedział ufności (prawa strona)**	**=B6/(1-B8)**									

Przykład 4.

Oszacować na poziomie ufności (1 – α) = 0.99 procent braków w wyprodukowanej partii produktów. Do badania wylosowana została próbka n=100 sztuk w której wykryto 16 produktów wadliwych.

	A	B	C	D	E	F	G	H	I	J	K
1	**Rozmiar próby n**	100									
2	**Ilość wadliwych**	16									
3											
4	**poziom ufności (1 – α)**	0,99									
5	α	**=1-B4**									
6	ε	**=-ROZKŁAD.NORMALNY.S.ODW(B5/2)**									
7	**p**	**=B2/B1**									
8	**d=ε√(p(1-p)/n**	**=B6*PIERWIASTEK(B7*(1-B7)/B1)**									
9	**przedział ufności (lewa strona)**	**=B7-B8**									
10	**przedział ufności (prawa strona)**	**=B7+B8**									

3.3. Minimalna liczebność próby

Lp.	Szacowanie	Uwagi	Minimalna liczebność	Parametry
1.	μ	znane σ	$n=\left(\varepsilon\frac{\sigma}{d}\right)^2$	$\varepsilon=-\Phi^{-1}\left(\frac{\alpha}{2}\right)$
2.	μ	nieznane σ duża próba	*jak w 1.*	$\sigma \approx S$
3.	μ	nieznane σ mała próba	$n=\left(\varepsilon\frac{s}{d}\right)^2$	$\varepsilon=-F^{-1}_{t(n-1)}\left(\frac{\alpha}{2}\right)$
4.	μ	metoda Steina nieznane σ mała próba	*jak w 3.* *jeśli* $n_0<n$ to *dolosować* $(n-n_0)$	ε *jak w 3.*
5.	p	*można oszacować* $\hat{p}$	$n=\frac{\varepsilon^2\,\hat{p}(1-\hat{p})}{d^2}$	$\hat{p}=\frac{m}{n}$, ε *jak w 1.*
6.	p	*nie można oszacować* $\hat{p}$, *przyjmuje się:* $\hat{p}=0.5$	$n=\frac{\varepsilon^2}{4d^2}$	ε *jak w 1.*

Przykład 5.

Na podstawie danych z przykładu 1. wyznaczyć minimalną liczebność próby tak, żeby oszacować rzeczywistą długość detalu z dokładnością 1.

	A	B	C	D	E	F	G	H	I	J	K
1	**poziom ufności (1 – α)**	0,99									
2	**odchylenie standardowe σ**	1,5									
3	**maksymalny błąd szacunku d**	1									
4	α	**=1-B1**									
5	ε	**=-ROZKŁAD.NORMALNY.S.ODW(B4/2)**									
6	**n=(εσ/d)^2**	**=ZAOKR.W.GÓRĘ(POTĘGA(B5*B2/B3;2);1)**									

Przykład 6.

Ile niezależnych pomiarów należy wykonać aby na poziomie ufności $(1 - \alpha) = 0.99$ oszacować rzeczywistą długość detalu z błędem maksymalnym równym 1 jeżeli wstępna próba 10 niezależnych pomiarów dała wyniki jak w przykładzie 1?

	A	B	C	D	E	F	G	H	I	J	K
1	**Pomiary**	18	21	22,4	23	21,3	21,9	17,6	21	17,8	19,4
2	**Rozmiar próby n_0**	10									
3											
4	**poziom ufności $(1 - \alpha)$**	0,99									
5	**maksymalny błąd szacunku d**	1									
6	α	**=1-B4**									
7	**s**	**=ODCH.STANDARDOWE(B1:K1)**									
8	ε	**=ROZKŁAD.T.ODW(B6;B2-1)**									
9	**n=(εs/d)^2**	**=ZAOKR.W.GÓRĘ(POTĘGA(B8*B7/B5;2);1)**									

Przykład 7.

Ile produktów należy wylosować aby na poziomie ufności $(1 - \alpha) = 0.9$ z błędem maksymalnym równym 5% oszacować procent braków w wyprodukowanej partii produktów (których jest przypuszczalnie 10%).

	A	B	C	D	E	F	G	H	I	J	K
1	**poziom ufności $(1 - \alpha)$**	0,9									
2	**maksymalny błąd szacunku d**	0,05									
3	**p**	0,1									
4											
5	α	**=1-B1**									
6	ε	**=-ROZKŁAD.NORMALNY.S.ODW(B5/2)**									
7	**$n=\varepsilon^2$ p (1-p) / d^2**	**=ZAOKR.W.GÓRĘ(POTĘGA(B6;2)*B3*(1-B3)/POTĘGA(B2;2);1)**									